

A Computationally-Discovered Simplification of the Ontological Argument

Paul E. Oppenheimer
CSLI/Stanford University

and

Edward N. Zalta
CSLI/Stanford University*

Abstract

The authors investigate the ontological argument computationally. The premises and conclusion of the argument are represented in the syntax understood by the automated reasoning engine PROVER9. Using the logic of definite descriptions, the authors developed a valid representation of the argument that required three non-logical premises. PROVER9, however, discovered a simpler valid argument for God’s existence from a single non-logical premise. Reducing the argument to one non-logical premise brings the investigation of the soundness of the argument into better focus. Also, the simpler representation of the argument brings out clearly how the ontological argument constitutes an early example of a ‘diagonal argument’ and, moreover, one used to establish a positive conclusion rather than a paradox.

*This paper is published in the *Australasian Journal of Philosophy*, 89 (2) (June 2011): 333–349. The authors would like to thank Branden Fitelson for introducing us to OTTER, PROVER9, and MACE4, and Bill McCune for answering our questions about them. We’d also like to thank Johan van Benthem, Nathan Tawil, and Jeffrey Kegler for comments about the paper, which led us to develop additional interesting points about the PROVER9 proof.

1. Introduction

Anselm’s ontological argument has come in for criticism ever since it was first proposed. But we think that the focus on finding flaws in the argument may have hindered progress in logically representing the argument in its most elegant form. We hope to show that computational techniques offer a new insight into Anselm’s ontological argument and demonstrate that there is much beauty inherent in its logic.

We take as our starting point the formulation in Oppenheimer and Zalta [1991]. This paper described in detail the logical axioms and non-logical premises needed to validly argue God’s existence in the manner of Anselm’s *Proslogion*. The analysis revealed that the logic of definite descriptions and three non-logical premises (one of which places a weak condition on the *greater than* relation) not only justified Anselm’s introduction of the definite description “that (conceivable thing) than which nothing greater is conceivable” but also validly implied God’s existence. (We took no stand on the truth of the premises in that paper.)

Recently, however, we decided to investigate that 1991 analysis computationally, and represented the 1991 formulation of the argument in the automated reasoning system PROVER9.¹ Once the premises and conclusion of the argument were represented in the syntax understood by PROVER9, it subsequently discovered a *simpler* valid argument for God’s existence from a single non-logical premise. By reducing the argument to one non-logical premise, the question about the soundness of the argument comes into better focus. Moreover, the simpler representation of the argument not only brings out the beauty of the logic inherent in the argument, but also clearly shows how it constitutes an early example of a ‘diagonal argument’ used to establish a positive conclusion rather than a paradox. Our results, described in detail below, also serve to forward the new discipline of computational metaphysics, as described in Fitelson and Zalta [2007].

2. A Review of the Ontological Argument

In Oppenheimer and Zalta [1991], we formulated Anselm’s ontological argument in a predicate calculus having the following three distinguishing

¹See <<http://www.cs.unm.edu/~mccune/prover9/>>, which will eventually be replaced by <<http://www.prover9.org/>>. PROVER9 is the successor to the program OTTER. See McCune [2009], [2003a] and [2003b].

features:

- (1) the system includes primitive definite descriptions, governed by the classic Russell [1905] axiom,
- (2) the system employs free logic for definite descriptions, and
- (3) the system includes an existence predicate ‘ $E!x$ ’ (‘ x exists’) which is not defined in terms of a quantified formula of the form ‘ $\exists y\phi$ ’ (‘there is a y such that ϕ ’), and in particular, it is not defined as $\exists y(y=x)$.

We review these features briefly in turn.

Concerning (1), we introduced primitive terms of the form $ix\phi$, to be governed by the following Russellian axiom schema, where $\chi[\alpha/\beta]$ is the result of substituting α for β everywhere in χ :

Description Axiom:

$\psi[ix\phi/z] \equiv \exists y(\phi[y/x] \ \& \ \forall u(\phi[u/x] \rightarrow u=y) \ \& \ \psi[y/z])$, where ψ is an atomic formula or identity formula in which z occurs free.

Here is a simple instance of this axiom schema, where ψ is set to Rbz and ϕ is set to Gx :

$$RbixGx \equiv \exists y(Gy \ \& \ \forall u(Gu \rightarrow u=y) \ \& \ Rby)$$

This asserts that b bears R to $ixGx$ iff something y is such that: (a) y has the property G , (b) anything that has the property G just is y , and (c) b bears R to y .

From this classic version of Russell’s axiom schema governing descriptions, there are several interesting logical theorem schemata governing descriptions, all of which played a role in the 1991 formulation, where $\exists!x\phi$ (‘there is a unique x such that ϕ ’) has its usual definition as $\exists x(\phi \ \& \ \forall y(\phi[y/x] \rightarrow y=x))$:

Description Theorem 1: $\exists!x\phi \rightarrow \exists y(y=ix\phi)$

Lemma 1: $\tau=ix\phi \rightarrow \phi[\tau/x]$, for any term τ

Description Theorem 2: $\exists y(y=ix\phi) \rightarrow \phi[ix\phi/x]$

Description Theorem 1 asserts: if there is a unique x such that ϕ , then there is something that is *the* x such that ϕ . Semantically, this tells

us that if there is a unique object in the domain satisfying ϕ , then the definite description, $ix\phi$, is well-defined (i.e., has a denotation). A typical instance of Lemma 1 might be $y=ixFx \rightarrow Fy$, which asserts that if the object y is identical to the x that is F , then y is F . Lemma 1 can then be used to prove Description Theorem 2, which asserts: if there is something that is the x such that ϕ , then *it* is such that ϕ . Intuitively, this tells us that well-defined definite descriptions $ix\phi$ can be substituted for the free variable x everywhere inside ϕ . The derivations of all three logical theorems appear in Oppenheimer and Zalta [1991].

Concerning (2), we used a free logic for definite descriptions so that it would be clear that definite descriptions can’t be substituted into universal claims without first knowing that they are well-defined. In free logic, the axiom schema underlying the rule of Universal Elimination ($\forall E$) is:

$$\forall x\phi \rightarrow (\exists y y=\tau \rightarrow \phi[\tau/x]), \text{ where } \tau \text{ is a constant or description}$$

So in free logic, the rule $\forall E$ allows us to instantiate the description $ix\phi$ into a universal claim only when we know $\exists y(y=ix\phi)$. Thus, it prevents one from substituting *arbitrary* descriptions into universal claims, an inference which may lead from truth to falsehood when the description is not well-defined. For example, a non-denoting description, say $ixFx$, may not be substituted into a true universal claim, say $\forall xPx$, to infer the formula $PixFx$, for the latter would be false when the description fails to denote (given a classical semantics for the description). By using free logic, we are prevented from substituting the description ‘that than which nothing greater can be conceived’ into universal claims without first establishing that there is something which is the thing than which nothing greater can be conceived.

Note also that in free logic, the following two axioms (the second is an axiom schema) are logical truths (i.e., true in every classical interpretation of the language):

$$\exists y(y=x)$$

$$\psi(\kappa) \rightarrow \exists y(y=\kappa), \text{ where } \kappa \text{ is any constant, and } \psi(\kappa) \text{ is any atomic or identity formula containing an occurrence of } \kappa$$

Clearly, the first of these is a logical truth. In any interpretation of the language, we assume the domain is classical and thus contains at least one object. So in any assignment to the variables, ‘ x ’ can be assigned

to an object in the domain. This is sufficient for the quantified identity statement $\exists y(y = x)$ to be true. The second is also a logical truth: for an atomic or identity formula ψ to be true in an interpretation, each of the terms in ψ must have a denotation. That is, each term in ψ must be well-defined. So the consequent $\exists y(y = \kappa)$ (for each term κ in ψ) is true. We shall see later in the paper that if we let κ be a definite description, then this claim can be derived. (See Description Theorem 3 in Section 3.4.)

Concerning (3), the system we formulated does not define $E!x$ (' x exists') as $\exists y(y = x)$. Semantically, the resulting system allows one to assert that only some objects in the domain of quantification have the property of existing. This proves useful in representing Anselm's use of language, for he presupposes that an object can exist in the understanding without also existing in reality, and he allows that some objects both exist in the understanding and exist in reality. We represent his claims about what exists in the understanding using formulas of the form $\exists x\phi$, and represent claims about what exists in reality using formulas of the form $\exists x(E!x \ \& \ \phi)$.

Having formulated a predicate calculus exhibiting features (1), (2), and (3), we then examined the non-logical elements of Anselm's ontological argument. We used two non-logical predicates, ' Cx ' (' x is conceivable') and ' Gxy ' (' x is greater than y ') to formulate the three non-logical premises needed in the argument. The first two non-logical premises needed are:

Connectedness of Greater Than: $\forall x\forall y(Gxy \vee Gyx \vee x=y)$

Premise 1: $\exists x(Cx \ \& \ \neg\exists y(Gyx \ \& \ Cy))$

We discuss these in turn.

The connectedness of the *greater than* relation asserts essentially that for every distinct pair of objects x, y in the domain of quantification, either Gxy or Gyx . Premise 1 asserts that there is a conceivable object such that nothing greater is conceivable. To simplify our discussion of Premise 1, we use ϕ_1 to abbreviate $Cx \ \& \ \neg\exists y(Gyx \ \& \ Cy)$. Thus, Premise 1 may be represented as $\exists x\phi_1$. In our 1991 paper, we showed the following interesting fact: the connectedness of the *greater than* relation suffices to establish the following important lemma:

Lemma 2: $\exists x\phi_1 \rightarrow \exists!x\phi_1$

Lemma 2 asserts that if there is a conceivable thing such that nothing greater can be conceived, then there is a unique conceivable thing such that nothing greater can be conceived. The proof of Lemma 2 was provided in our paper.² As we shall see below, Lemma 2 helps to justify the introduction of the description "that (conceivable thing) than which nothing greater is conceivable" ($\iota x\phi_1$) into the ontological argument.

The final non-logical premise needed in Anselm's argument is:

Premise 2: $\neg E!\iota x\phi_1 \rightarrow \exists y(Gy\iota x\phi_1 \ \& \ Cy)$

Premise 2 asserts: if that than which nothing greater is conceivable fails to exist, something greater than it is conceivable.

Finally, we defined 'God' (' g ') to be *the* conceivable thing such that nothing greater is conceivable:

$g =_{df} \ \iota x\phi_1$

With these premises, theorems, and definitions we then formulated Anselm's ontological argument as follows:

²The proof sketched in 1991 can be formalized as follows:

1.	$\exists x(Cx \ \& \ \neg\exists y(Gyx \ \& \ Cy))$	Assume antecedent.
2.	$Ca \ \& \ \neg\exists y(Gya \ \& \ Cy)$	from (1), by $\exists E$, ' a ' arbitrary
3.	$\exists z(z \neq a \ \& \ Cz \ \& \ \neg\exists y(Gyz \ \& \ Cy))$	Assumption, for <i>Reductio</i> .
4.	$b \neq a \ \& \ Cb \ \& \ \neg\exists y(Gyb \ \& \ Cy)$	from (3), by $\exists E$, ' b ' arbitrary
5.	$Gab \ \vee \ Gba \ \vee \ a=b$	Connectedness of Greater Than
6.	$Gab \ \vee \ Gba$	from (4) and (5), by $\vee E$
7.	Gab	Assumption
8.	$Gab \ \& \ Ca$	from (2) and (7), by $\&I$
9.	$\exists y(Gyb \ \& \ Cy)$	from (8), by $\exists I$
10.	$\neg Gab$	from (4), (7), and (9), by <i>Reductio</i>
11.	Gba	from (6) and (10), by $\vee E$
12.	$Gba \ \& \ Cb$	from (4) and (11), by $\&I$
13.	$\exists y(Gya \ \& \ Cy)$	from (12), by $\exists I$
14.	$\neg\exists z(z \neq a \ \& \ Cz \ \& \ \neg\exists y(Gyz \ \& \ Cy))$	from (2), (3), and (13), by <i>Reductio</i>

The only non-logical inference in this proof is at line 5, which cites the connectedness of the *greater than* relation.

- | | | |
|----|--|--|
| 1. | $\exists x\phi_1$ | Premise 1 |
| 2. | $\exists!x\phi_1$ | from (1), by Lemma 2 |
| 3. | $\exists y(y=ix\phi_1)$ | from (2), by Description Thm 1 |
| 4. | $Cix\phi_1 \ \& \ \neg\exists y(Gyix\phi_1 \ \& \ Cy)$ | from (3), by Description Thm 2 |
| 5. | $\neg E!ix\phi_1$ | Assumption, for <i>Reductio</i> |
| 6. | $\exists y(Gyix\phi_1 \ \& \ Cy)$ | from (5), by Premise 2 |
| 7. | $\neg\exists y(Gyix\phi_1 \ \& \ Cy)$ | from (4), by &E |
| 8. | $E!ix\phi_1$ | from (5), (6), and (7), by <i>Reductio</i> |
| 9. | $E!g$ | from (8), by the definition of g |

Note that, strictly speaking, we need not have used free logic to reconstruct the argument because we establish at line (3) that our description $ix\phi_1$ is well-defined. Its logic is therefore classical, and the ontological argument proceeds along classical lines.³

Clearly then, the ontological argument directly rests on the logical theorems Description Theorems 1 and 2 and on the non-logical premises Premise 1, Lemma 2, and Premise 2. It indirectly rests on the Description Axiom (which yields both Description Theorem 1 and Lemma 1, the latter yielding Description Theorem 2), and on the connectedness of the *greater than* relation (which together with Premise 1, yields Lemma 2).

3. The Computational Implementation

Following the example of Fitelson and Zalta [2007], we investigated the above analysis with the help of automated reasoning technologies. We implemented our 1991 formulation of the ontological argument in PROVER9, which is a well-known and easy-to-use theorem-proving environment. In what follows, we use `typewriter font` to indicate formulas that are in PROVER9 syntax. Our intention was to follow, as closely as possible, the structure of the argument in our 1991 paper, as outlined immediately above, when representing the premises and conclusion in PROVER9 syntax. However, first-order automated reasoning engines provide rather austere logical environments, and we had to employ some interesting techniques to represent the premises faithfully and to stay close to the structure of the argument. In this section, we first discuss the representation of the logical axioms and theorems used in the argument (Section 3.1), then

³We noted in our earlier paper that we used a free logic for descriptions mainly for psychological reasons, to forestall any concern that the argument smuggles in God's existence when Anselm starts *using* the definite description.

focus on the representations of the non-logical premises (Section 3.2), describe the valid argument discovered by PROVER9 using its own notation (Section 3.3), and finally, reverse engineer that argument into a more human-friendly logical form (Section 3.4).

3.1 Representing the Logic

To implement the inferences used in the ontological argument, we must at the very least try to represent Description Theorems 1 and 2 using PROVER9 syntax, since these were used in the main argument described above. However, a more complete implementation would represent Russell's Description Axiom and show how this implies Description Theorem 1 and Lemma 1, and subsequently, Description Theorem 2, in PROVER9 syntax. But for reasons of space, we shall not discuss all the details concerning these derivations in what follows.⁴ Our goal in this subsection is simply to show how these logical axioms and theorems are to be represented in PROVER9 syntax.

Unfortunately, PROVER9, like other first-order automated reasoning engines, (i) doesn't include primitive descriptions as syntactic terms, and (ii) can't represent axiom and theorem *schemata*, much less discover proofs of the latter. How then are we to represent the Description Axiom and the logic consisting of Description Theorem 1, Lemma 1, and Description Theorem 2, all of which are schemata including primitive descriptions?

There is probably no unique correct answer to this question, for there may be various ways to represent the needed logical theorems. We addressed problem (i) by introducing, and axiomatizing, a relational formula `Is_the(x,F)` to assert that x is the F . This works better than what appears to be the more natural method of introducing a term `the(F)` and asserting identities of the form $x = \text{the}(F)$.⁵ We addressed problem (ii) by first reconceiving the logical axiom schemata and logical theorem schemata as second-order quantifications and then emulating these second-order claims in PROVER9 by using multi-sorted first-order logic.

⁴The complete details can be examined on the webpage we have developed in support of this paper; see <https://mally.stanford.edu/cm/ontological-argument/>.

⁵With help from Christopher Menzel, we've determined that the use of formulas like $x = \text{the}(F)$ in PROVER9 leads to sorting irregularities. But we omit discussion of this technical issue here. The webpage <https://mally.stanford.edu/cm/ontological-argument/sorting-issue.html> describes the issue in detail.

Consequently, we introduced sortal predicates such as `Object`, `Relation1`, and `Relation2`, and we introduced formulas such as `Ex1(F, x)` and `Ex2(F, x, y)` to represent, respectively, the predications x exemplifies (the 1-place relation, i.e., property) F , and x and y exemplify (the 2-place relation) F .

Using these techniques, we looked at the simplest class of instances of Russell's Description Axiom that would be needed in the argument. The simplest class of instances emerges by setting ψ to Gz and ϕ to Fx , to yield instances of the form:

$$GxFx \equiv \exists y(Fy \ \& \ \forall u(Fu \rightarrow u=y) \ \& \ Gy)$$

In PROVER9 syntax, the left condition has to be represented by way of a conjunction, and the full representation of the above formula, including the suppressed quantifiers on the free relation variables, is as follows:

```
all F all G all x ((Relation1(F) & Relation1(G) & Object(x)) ->
  ((Is_the(x,F) & Ex1(G,x)) <-> (exists y (Object(y) & Ex1(F,y) &
    (all u ((Object(u) & Ex1(F,u)) -> u=y)) & Ex1(G,y)))))).
```

This asserts that for any 1-place relations F and G and object x , if x is the F and x exemplifies G , then there is an object y such that (a) y exemplifies F , (b) anything u that exemplifies F is identical to y , and (c) y exemplifies G . Clearly, this doesn't capture the full generality of the Description Axiom schema, but it does capture the significance of the class of instances that emerge when the property G is predicated of the F .

Similarly, we represented Description Theorem 1 in PROVER9 syntax as follows:

```
all F (Relation1(F) -> ((exists x (Object(x) & Ex1(F,x) &
  (all y (Object(y) -> (Ex1(F,y) -> y=x)))) ->
  (exists y (Object(y) & Is_the(y,F)))))).
```

This asserts that for any 1-place relation F , if there is an object x such that (a) x is F , and (b) any object y that exemplifies F is identical to x , then there is an object y which is the F .

If PROVER9 is given an input file consisting of the above representation of the Description Axiom as premise, and the above representation of Description Theorem 1 as conclusion, it finds a proof of the latter.⁶

⁶One can verify this by installing PROVER9 and inputting the file located at <<https://mally.stanford.edu/cm/ontological-argument/descripthm1.in>>.

The proof is shown to be non-trivial because the model-building program MACE4, which is included in the installation of PROVER9, shows that there is a model of the premise alone. This establishes (by the soundness theorem of first-order logic) that the premises are consistent and, therefore, that we don't have a trivial proof of Description Theorem 1 from an inconsistent set of premises.

The next theorem we thought was required for the ontological argument is Lemma 1, since it is used in the proof of Description Theorem 2. Lemma 1 was translated into PROVER9 syntax as follows:

```
all x all F all y ((Object(x) & Relation1(F) & Object(y)) ->
  ((Is_the(x,F) & x=y) -> Ex1(F,y))).
```

Of course, this might be simplified so that the consequent of the main conditional reads `Is_the(x,F) -> Ex1(F,x)`, but we thought that this alternative wouldn't capture the fact that the antecedent of Lemma 1 is an identity claim. We should note here that the representation of Lemma 1 must be derived from a representation of the Description Axiom that expresses a different class of instances than that discussed earlier.⁷

Recall next that our analysis also showed that the ontological argument requires Description Theorem 2. We represented this theorem in PROVER9 syntax as:

```
all F (Relation1(F) ->
  ((exists y (Object(y) & Is_the(y,F))) ->
  (all z (Object(z) -> (Is_the(z,F) -> Ex1(F,z)))))).
```

In other words, if something is the F , then anything that is the F exemplifies F . PROVER9 derives this from the representation of Lemma 1.⁸

It is clear from the above representations of the logical axioms and theorems governing descriptions that instead of introducing sorted variables, we are using restricted quantification over the sorts `Object`, `Relation1`, and `Relation2`. Sometimes, the relations among these sorts must be explicitly formulated as additional premises, called sorting principles.

⁷If we set ψ to $z=w$ and ϕ to Gx , then the following captures a class of instances of the Description Axiom:

$$x F x = w \equiv \exists y(Fy \ \& \ \forall u(Fu \rightarrow u=y) \ \& \ y=w)$$

Once this is represented in PROVER9 syntax, PROVER9 can derive the representation of Lemma 1. See <<https://mally.stanford.edu/cm/ontological-argument/lemma1.in>>.

⁸See <<https://mally.stanford.edu/cm/ontological-argument/descripthm2.in>>.

PROVER9 requires that we introduce a sorting principle to govern the formula $\text{Is_the}(x,F)$ that represents definite descriptions. This principle asserts that the entity that satisfies the description must be an **Object**, and the matrix of the description must be a **Relation1**:

```
all x all F (Is_the(x,F) -> (Relation1(F) & Object(x))).
```

We mention this because the above formula played a role in the simplified version of the ontological argument that PROVER9 discovered.

Thus, even though PROVER9 does independently establish that the representation of the Description Axiom implies the representations of Description Theorem 1 and Lemma 1, and that the latter implies the representation of Description Theorem 2, we simplified the actual input file that implements the ontological argument computationally: we included only the representations of Description Theorems 1 and 2 and the above sorting principle, since these were the logical premises needed for the main argument.

3.2 Representing the Non-logical Principles

The connectedness of the *greater than* relation is easily represented in PROVER9 syntax:

```
all x all y ((Object(x) & Object(y)) ->
  (Ex2(greater_than,x,y) | Ex2(greater_than,y,x) | x=y)).
```

Next, our representation of Premise 1 employs a definition of the property *none_greater*. We introduced this definition because the reasoning in the argument often treats the complex formula ϕ_1 as a unit. If we define a new 1-place predicate **none_greater** in PROVER9 syntax, that predicate becomes substitutable for the variable **F** when it is restricted to the sort **Relation1**. So, to represent Premise 1, we define: x is *none_greater* iff x is conceivable and it is not the case that there is an object greater than x that is conceivable. In PROVER9 syntax:

```
all x (Object(x) -> (Ex1(none_greater,x) <->
  (Ex1(conceivable,x) &
  -(exists y (Object(y) & Ex2(greater_than,y,x) &
  Ex1(conceivable,y)))))).
```

Thus, the predicate **none_greater** is the counterpart of the abbreviated open formula ϕ_1 . We may then represent Premise 1 in PROVER9 syntax very simply, as follows:

```
exists x (Object(x) & Ex1(none_greater,x)).
```

We turn next to another non-logical claim that we thought was needed to help justify the introduction of the description $\iota x\phi_1$. Lemma 2 tells us that if there is something than which none greater is conceivable, then there is a unique thing than which none greater is conceivable. Lemma 2 is part of the justification for the introduction of the description $\iota x\phi_1$ when given its antecedent (i.e., Premise 1). Lemma 2 can be represented in PROVER9 syntax as:

```
exists x (Object(x) & Ex1(none_greater,x)) ->
  exists x (Object(x) & Ex1(none_greater,x) &
  (all y (Object(y) -> (Ex1(none_greater,y) -> y=x)))).
```

If one gives PROVER9 a file containing as premises the above representations of the connectedness of *greater than* and of the definition of *none_greater*, and Lemma 2 as conclusion, PROVER9 will find a valid proof.⁹

Recall next that Premise 2 is: if the conceivable thing than which none greater is conceivable fails to exist, then something greater than it is conceivable. We represented this premise in PROVER9 syntax as:

```
all x (Object(x) -> ((Is_the(x,none_greater) & -Ex1(e,x)) ->
  exists y (Object(y) & Ex2(greater_than,y,x) &
  Ex1(conceivable,y)))).
```

Here the letter ‘e’ is the existence predicate.

Finally, the definition of ‘God’ as *the* conceivable thing x than which none greater is conceivable is represented in PROVER9 syntax as:

```
Is_the(g,none_greater).
```

Thus, the representation of the conclusion of the ontological argument, that God exemplifies existence, is simply: $\text{Ex1}(e,g)$.

When we input the above representations of the logical and non-logical premises to PROVER9, we thought it would prove God’s existence from the

⁹See <<https://mally.stanford.edu/cm/ontological-argument/lemma2.in>>.

following non-logical premises: the definition of *none_greater*, Premise 1, Lemma 2, Premise 2, and the definition of God (with the connectedness of *greater than* being used in a separate proof of Lemma 2). However, we were surprised to find that PROVER9 did not require all of these non-logical elements.

3.3 PROVER9's Ontological Argument

When PROVER9 accepts a set of first-order formulas as input (e.g., our representations above), it begins its proof search by reformulating the content of these formulas into clausal normal form. In clausal normal form: (1) each clause is a finite disjunction of *literals*, where a literal is either an atomic formula or a negation of an atomic formula (and since only atomic formulas can be negated, clausifying a first-order formula involving a conjunction or biconditional results in a list of clauses), and (2) the universally quantified variables in the first-order formulas are replaced by free variables and the existentially quantified variables by Skolem functions (Skolem [1920]).¹⁰ Once PROVER9 clausifies the premises, it then

¹⁰To fully understand how PROVER9 works, the reader should examine how it clausifies a complex formula involving conjunctions, biconditionals, and existential quantifiers, such as the definition of the *none_greater* predicate given earlier in the text. The clausification consists of the following list of five clauses:

```
-Object(x) | -Ex1(none_greater,x) | Ex1(conceivable,x).
-Object(x) | -Ex1(none_greater,x) | -Object(y) | -Ex2(greater_than,y,x) | -Ex1(conceivable,y).
-Object(x) | Ex1(none_greater,x) | -Ex1(conceivable,x) | Object(f3(x)).
-Object(x) | Ex1(none_greater,x) | -Ex1(conceivable,x) | Ex2(greater_than,f3(x),x).
-Object(x) | Ex1(none_greater,x) | -Ex1(conceivable,x) | Ex1(conceivable,f3(x)).
```

The definition given in the main text is a universal claim over a conditional whose antecedent is *Object(x)*. PROVER9 eliminates the quantifier “all *x*” and then turns the conditional into a disjunction, preserving the negated antecedent of the conditional as the first disjunct in every one of the five clauses above. PROVER9 then processes the biconditional having *Ex1(none_greater,x)* as the left condition (the definiendum). The left-to-right direction of this biconditional is turned into the first two clauses above, while the right-to-left direction is turned into the remaining three clauses. Though the Skolem function *f3(x)* is used in the final three clauses to replace the existential quantifier in the right-to-left direction of the biconditional, it isn't needed in the first two clauses that result from the left-to-right direction since PROVER9 can define away the negated existential statement in terms of a universal statement.

As another example, see <<https://mally.stanford.edu/cm/ontological-argument/clausifyingDescThm1.html>> for a step-by-step account of how Description Theorem 1 gets converted into a group of six clauses (in clausal normal form) by PROVER9 during the preprocessing stage.

clausifies the negation of the conclusion, takes that as another premise, and then attempts to derive a contradiction from the resulting premise set.¹¹ It is important to note that when PROVER9 finds a proof of a conclusion from some premises, it does not always appeal to all of the clauses from all of the premises in the proof.

PROVER9 easily discovers a proof of the claim that God exists from the above representations. However, much to our surprise, the proof it discovered used only a few of the premises we formulated above. PROVER9 reports that it used only the following premises in the proof:

```
all F (Relation1(F) -> ((exists x (Object(x) & Is_the(x,F))) ->
  (all y (Object(y) -> (Is_the(y,F) -> Ex1(F,y)))))).

all x all F (Is_the(x,F) -> (Relation1(F) & Object(x))).

all x (Object(x) -> (Ex1(none_greater,x) <->
  (Ex1(conceivable,x) & -(exists y (Object(y) &
  Ex2(greater_than,y,x) & Ex1(conceivable,y)))))).

all x (Object(x) -> ((Is_the(x,none_greater) & -Ex1(e,x)) ->
  (exists y (Object(y) & Ex2(greater_than,y,x) &
  Ex1(conceivable,y)))))).

Is_the(g,none_greater).
```

As you can see by inspection, the first of these is Description Theorem 2, the second is the sorting principle on *Is_the*, the third is the definition of *none_greater*, the fourth is Premise 2, and the last is the definition of ‘*g*’. PROVER9 did *not* need to use Description Theorem 1, Premise 1, or Lemma 2. Moreover, since it didn't use Lemma 2, it didn't require the connectedness of *greater than*. Indeed, PROVER9 didn't need the full content of the premises that it did use in its proof; each of these premises gets turned into multiple clauses and only some of the resulting clauses are used in the proof.

¹¹Those interested in seeing PROVER9's clausifications of all of the representations described above may inspect <<https://mally.stanford.edu/cm/ontological-argument/clauses.html>>.

3.4 The Simplified Ontological Argument

We will not here reproduce the proof that PROVER9 discovered, since proofs in clausal normal form are somewhat difficult to read. We make it available as an output file.¹² Instead, we report on our *analysis* of the reasoning PROVER9 used in its proof, and we reconstruct, in standard logical notation, a new version of the ontological argument based on the one PROVER9 constructed.

PROVER9 simplified the proof by employing the equivalent of the following logical theorem schema, Description Theorem 3, which is derivable from the Description Axiom.¹³

Description Theorem 3: $\psi[\iota x\phi/z] \rightarrow \exists y(y = \iota x\phi)$, where ψ is any atomic or identity formula with z free.

In other words, if the description the x such that ϕ appears in a true atomic or identity formula then there is something which is the x such that ϕ . Semantically speaking, this tells us that if an atomic or identity formula containing a description is true, the description must have a denotation.

Using Description Theorems 2 and 3, we can now establish that $E!\iota x\phi_1$ from a single non-logical premise, namely Premise 2. Once $E!\iota x\phi_1$ is

¹²See <<https://mally.stanford.edu/cm/ontological-argument/ontological.out>>. Alternatively, the reader may install PROVER9 and use our input file <<https://mally.stanford.edu/cm/ontological-argument/ontological.in>>.

¹³To see this, assume the antecedent; i.e., suppose $\psi[\iota x\phi/z]$, to show $\exists y(y = \iota x\phi)$. Then by the Description Axiom,

$$\exists y(\phi[y/x] \ \& \ \forall u(\phi[u/x] \rightarrow u=y) \ \& \ \psi[y/z])$$

Call an arbitrary such object ‘ b ’; so we know

$$\phi[b/x] \ \& \ \forall u(\phi[u/x] \rightarrow u = b) \ \& \ \psi[b/z]$$

However, we also know that the following is an instance of the Description Axiom:

$$b = \iota x\phi \equiv \exists y(\phi[y/x] \ \& \ \forall u(\phi[u/x] \rightarrow u = y) \ \& \ b = y)$$

So if we can show the right side of this biconditional, we are done. But $b = b$, and we already know $\phi[b/x] \ \& \ \forall u(\phi[u/x] \rightarrow u = b)$. So, conjoining, we know:

$$\phi[b/x] \ \& \ \forall u(\phi[u/x] \rightarrow u = b) \ \& \ b = b$$

from which it follows, by generalizing on ‘ b ’:

$$\exists y(\phi[y/x] \ \& \ \forall u(\phi[u/x] \rightarrow u = y) \ \& \ b = y)$$

So, using the instance of the Description Axiom displayed above, we may infer $b = \iota x\phi$, and thus $\exists y(y = \iota x\phi)$.

established, it follows that the definition of ‘God’ abbreviates a well-defined description, since the description appears in a true atomic formula. The resulting simplified ontological argument for the existence of God is:

- | | | |
|----|--|--|
| 1. | $\neg E!\iota x\phi_1$ | Assumption, for <i>Reductio</i> |
| 2. | $\exists y(Gy\iota x\phi_1 \ \& \ Cy)$ | from (1), by Premise 2 and MP |
| 3. | $Gh\iota x\phi_1 \ \& \ Ch$ | from (2), by $\exists E$, ‘ h ’ arbitrary |
| 4. | $Gh\iota x\phi_1$ | from (3), by $\&E$ |
| 5. | $\exists y(y = \iota x\phi_1)$ | from (4), by Desc. Thm. 3 |
| 6. | $C\iota x\phi_1 \ \& \ \neg\exists y(Gy\iota x\phi_1 \ \& \ Cy)$ | from (5), by Desc. Thm. 2 |
| 7. | $\neg\exists y(Gy\iota x\phi_1 \ \& \ Cy)$ | from (6), by $\&E$ |
| 8. | $E!\iota x\phi_1$ | from (1), (2), (7), by <i>Reductio</i> |
| 9. | $E!g$ | from (8), by the definition of ‘ g ’ |

In the context of a free logic, note that we can’t just move from (8) to (9) by definition. Strictly speaking, once we have arrived at (8), we must infer that the description denotes before we can substitute ‘ g ’ for the description. But, indeed, the claim that establishes that the description denotes, i.e., $\exists y(y = \iota x\phi_1)$, follows from (8) by Description Theorem 3. So we have established that the description $\iota x\phi_1$ is well-defined, and therefore that the introduced constant ‘ g ’ is also well-defined. Thus, by the principles of free logic, we may substitute the definiendum ‘ g ’ for the definiens $\iota x\phi_1$.

Thus, we have a valid argument for the existence of God that doesn’t require that any conditions be placed on the *greater than* relation, and doesn’t require that we assert Premise 1 or establish Lemma 2 in order to justify the introduction of the definite description $\iota x\phi_1$. Instead, all the reasoning about the description takes place inside a *Reductio* assumption, except at the very end, after it is established that the description is well-defined. The question of the soundness of the ontological argument now reduces to the question of the truth of Premise 2! We shall discuss the question of soundness in the final section of the paper, but first, we turn to some observations about the new representation of the ontological argument.

4. Observations

First, we think this analysis shows the value of employing computational techniques in the study of metaphysics. These techniques reveal not only that the ontological argument can be greatly simplified, but also that one can justify the introduction of the definite description $\iota x\phi_1$ implicitly,

within a *Reductio* environment, instead of arguing directly and explicitly that the description is well-defined. Our computational techniques reveal that minimal logical and non-logical machinery is necessary for formulating an ontological argument for the existence of God. Indeed, if the reader were to analyze which clauses of each premise are used in the PROVER9 proof, then it would become clear exactly what part of the content of each of the logical and non-logical premises is actually used in the argument. (We omit this clausal analysis here.) In any case, the computationally-simplified version of the argument reveals that it has a subtle logical beauty.

Second, it is important to note that the new, simpler argument doesn't do all of the things that Anselm's original argument does. In *Proslogion II*, it appears that Anselm starts to use the description "that than which nothing greater can be conceived" before he has established that it is well-defined. Our earlier 1991 formulation shows that he was justified in doing this, if given only the connectedness of the *greater than* relation, since Premise 1 is a claim that Anselm endorses.

Third, the use of the description $\iota x\phi_1$ in line (1) of the new, simpler proof doesn't need justification because the *Reductio* assumption, $\neg E!\iota x\phi_1$, doesn't assume that the description has a denotation. The *Reductio* assumption could be true for the reason that the description has no denotation, in which case the atomic formula $E!\iota x\phi_1$ is false (making the *Reductio* assumption true). The new version of the argument shows that the use of the description $\iota x\phi_1$ in an atomic predication becomes justified (given Premise 2), only starting at line (8). Moreover, the automated reasoning engine isolated a single premise which has the property that once you assume its antecedent, the consequent implies its own negation (note the argument from lines (2)–(7) in the version of the argument immediately above). Since the antecedent *is* the assumption used for *Reductio*, the premise contains a guarantee that the *Reductio* will succeed.

Fourth, it is interesting to note that one can (i) abandon the definition of God as $\iota x\phi_1$, (ii) generalize Premise 2 to the claim that $\neg E!x \rightarrow \exists y(Gyx \ \& \ Cy)$ and still (iii) develop a valid argument to the conclusion that anything that satisfies ϕ_1 exemplifies existence. For suppose some arbitrary object, say b , satisfies ϕ_1 , to show $E!b$. For *Reductio*, assume $\neg E!b$. Then by the generalized Premise 2, $\exists y(Gyb \ \& \ Cy)$. But this contradicts the second conjunct of the assumption that b satisfies ϕ_1 , i.e., the second conjunct of the claim that $Cb \ \& \ \neg\exists y(Gyb \ \& \ Cy)$. So $E!b$. Of

course, we can't justify Premise 2, as stated above, from this generalized version of Premise 2 ($\neg E!x \rightarrow \exists y(Gyx \ \& \ Cy)$). That's because you can't instantiate the description $\iota x\phi_1$ for x in the generalized version until you first prove $\exists y(y = \iota x\phi_1)$, given the restricted rule of universal elimination ($\forall E$) for the free logic of descriptions.

Fifth, the new analysis of the argument brings out much more clearly that it deploys diagonal reasoning for a positive conclusion. By contrast, most diagonal arguments in the history of philosophy have been deployed to develop paradoxes. Anselm diagonalizes when he applies the description to itself in line (6), i.e., when he invokes Description Theorem 2 after concluding within the *Reductio* that there is something which is *the* conceivable thing such that nothing greater is conceivable. Description Theorem 2 allows him to infer that the object denoted by the description satisfies the matrix of the description, i.e., that it is itself conceivable and such that nothing greater is conceivable. Since the description itself is instantiated within its own matrix, we have a clear case of diagonalization. But here the diagonal argument leads to an existence claim, rather than to a nonexistence claim as in Russell's Paradox. More generally, diagonal arguments have been used to reach negative claims, such as in Cantor's proofs that the power set of a set A can't be mapped 1-to-1 to a subset of A and that there is no 1-to-1 mapping from the set of real numbers to the set of natural numbers. Diagonal arguments have also been employed to generate *aporiai*, or puzzles such as the Liar Paradox.

Sixth, our new analysis offers an additional insight into how much one has to presuppose about the *greater than* relation to get the ontological argument off the ground. Prior to our 1991 paper, it was generally assumed that the *greater than* relation used in the ontological argument had to be an *ordering* of some kind. Our 1991 paper showed, however, that it suffices for the validity of the ontological argument that the *greater than* relation be a *connected* relation and satisfy Premises 1 and 2. Now, the present analysis shows that the *greater than* relation doesn't even have to be connected or satisfy Premise 1. It simply has to satisfy Premise 2, i.e., be such that $\neg E!\iota x\phi_1 \rightarrow \exists y(Gy\iota x\phi_1 \ \& \ Cy)$. We think it is striking that *greater than* need have so little content for the ontological argument to be valid.

Finally, it is worth mentioning that the computational tools we used include a model-building program. PROVER9 comes with the model-building program MACE4, and the latter comes in very handy. We fre-

quently used MACE4 to investigate whether a set of premises was consistent. MACE4 would then try to find the smallest model in which all the premises are true. On some occasions, when we set PROVER9 on a problem, to attempt to find a proof of a conclusion from some premises, PROVER9 would grind on and on. In such cases, we often used MACE4 to see whether it could find a countermodel in which the premises are true and the conclusion false; if it were to find one, we would know that the premise set was not yet strong enough to imply the conclusion. By examining such countermodels, we often could determine what the missing premises had to be. We strongly recommend that these computational methods be included in the study of metaphysics.

5. Concerning Soundness

One virtue of our new version of the ontological argument is that it shows Anselm could have proved God's existence with less metaphysics and more (non-modal) logic. He doesn't need to assert that there is something conceivable such that nothing greater can be conceived. He doesn't need the connectedness of the *greater than* relation. He doesn't need Lemma 2, nor does he need to justify the introduction of the definite description into the proof. Indeed, as we have argued elsewhere, Premise 1 is a premise that should be challenged (Oppenheimer and Zalta [2007]). When one abandons it as we have done in the above version of the argument, the only thing left to challenge is Premise 2.

Premise 2 is:

$$\neg E!x\phi_1 \rightarrow \exists y(Gyx\phi_1 \ \& \ Cy)$$

This reads as follows: if the conceivable thing than which nothing greater is conceivable fails to exist, then something greater than it is conceivable. This has some *prima facie* plausibility. There is no *de re/de dicto* ambiguity in this premise, given that it has already been formally represented. Moreover, it doesn't presuppose, in the technical sense of presupposition, that anything answers to the description. As we shall see, this conditional can be false without implying the existence of something such that nothing greater is conceivable. On one technical sense of 'presuppose', a formula ϕ presupposes ψ only if both ϕ implies ψ and $\neg\phi$ implies ψ . But Premise 2 doesn't presuppose $\exists y(y = ix\phi_1)$ in this sense, since the negation of Premise 2 doesn't imply $\exists y(y = ix\phi_1)$, as we shall see below. The

fact that Premise 2 doesn't presuppose that there is something such that nothing greater is conceivable undermines an objection one might make to Premise 2, namely, that it presupposes that the definite description has a denotation. So what, if anything, is wrong with Premise 2?

One can argue systematically against it as follows. Since Premise 2 is a conditional, in order to show that it is false one must argue both that the antecedent, $\neg E!x\phi_1$, is true and that the consequent, $\exists y(Gyx\phi_1 \ \& \ Cy)$, is false. However, there are two different ways for the antecedent of Premise 2 to be true: on the one hand, the description $ix\phi_1$ could fail to denote, in which case, the atomic formula $E!x\phi_1$ is false and its negation (the antecedent) true; on the other, the description does denote, and the object it denotes fails to have the property of existence. So one may argue against Premise 2 disjunctively: (1) Suppose $ix\phi_1$ fails to denote and the antecedent of Premise 2 is therefore true. If so, then the consequent is false, on the grounds that if the description fails to denote, then a claim of the form $Gyx\phi_1$ is false for every y (since it is an atomic formula with a non-denoting term). If $Gyx\phi_1$ is false for every y , then $Gyx\phi_1 \ \& \ Cy$ is false for every y . Therefore, the consequent of Premise 2 is false. (2) Suppose $ix\phi_1$ denotes and the antecedent of Premise 2 is true because the object denoted lacks existence. In such a situation, where there is a unique thing such that nothing greater can be conceived, the consequent of Premise 2 is false, since it is inconsistent with there being a unique thing such that nothing greater can be conceived.

Do arguments (1) and (2) establish that Premise 2 is false? Not quite. In the case in which the description $ix\phi_1$ denotes and the object it denotes exists, the antecedent of Premise 2 is false, making Premise 2 true. But the defender of the ontological argument can take no comfort from such an observation, since it defends Premise 2 by using the conclusion of the ontological argument. That is, if she uses the existence of the conceivable thing than which no greater thing is conceivable to prove Premise 2, she is guilty of circular reasoning. She needs an independent argument to support the premise. Thus, arguments (1) and (2) above show that the defender of the ontological argument needs independent support for two claims: that the definite description denotes and that Premise 2 is true. Our 1991 analysis of the argument is still relevant, since it shows how the ontological arguer could justify Anselm's use of the definite description.¹⁴ The present analysis shows why the use of the definite description

¹⁴Given the argument outlined above against Premise 2, a defender of Anselm might

needs independent justification. Consequently, though the simplified ontological argument is valid, Premise 2 is questionable and to the extent that it lacks independent justification, the simplified argument fails to demonstrate that God exists. The use of computational techniques in systematic metaphysics has illuminated the relationship between Premise 2 of the ontological argument and the conclusion that God exists.

References

Anselm, St., *Proslogion II*, (J. Barnes, tr.), in Barnes 1972.

Barnes, Jonathan, 1972, *The Ontological Argument*, London: Macmillan.

Fitelson, Branden, and Edward N. Zalta, 2007, 'Steps Towards a Computational Metaphysics', *Journal of Philosophical Logic*, 36/2: 227–247.

McCune, William, 2003a, *Otter 3.3 Reference Manual* (Tech. Memo ANL/MCS-TM-263, Mathematics and Computer Science Division), Argonne, IL: Argonne National Laboratory.

McCune, William, 2003b, *Mace4 Reference Manual and Guide* (Tech. Memo ANL/MCS-TM-264, Mathematics and Computer Science Division), Argonne, IL: Argonne National Laboratory.

McCune, William, 2009, *Prover9 Manual*, URL = <http://www.cs.unm.edu/~mccune/mace4/manual/2009-11A/>.

Oppenheimer, Paul E., and Edward N. Zalta, 1991, 'On the Logic of the Ontological Argument', *Philosophical Perspectives*, 5: 509–529; reprinted in *The Philosopher's Annual: 1991*, Volume XIV (1993): 255–275.

Oppenheimer, Paul E., and Edward N. Zalta, 2007, 'Reflections on the Logic of the Ontological Argument', *Studia Neoscholastica*, 4/1: 28–35.

consider whether the ontological argument can be strengthened by using our original formulation as in 1991, but with the *general* form of Premise 2 discussed earlier: $\neg Ex \rightarrow \exists y(Gyx \ \& \ Cy)$. The justification of this more general premise may not be subject to the same circularity that infects the justification of Premise 2 (though, of course, it may have problems of its own).

Russell, Bertrand, 1905, 'On Denoting', *Mind* (new series), 14: 479–493.

Skolem, Thoralf, 1920, 'Logisch-kombinatorische Untersuchungen über die Erfüllbarkeit oder Beweisbarkeit mathematischer Sätze nebst einem Theoreme über dichte Mengen', *Videnskapsselskapet Skrifter, I. Matematisk-naturvidenskabelig Klasse*, 6: 1–36.